

SCIENCE & VIE

MENSUEL - N° 879
DÉCEMBRE 1990

● Les autoradios
au banc d'essai

**DROGUE :
LE PLAISIR
QUI TUE**

M 2578 - 879 - 20,00 F



● Critique
des livres
scolaires

Jouons au Puissance 4

INFORMATIQUE AMUSANTE

Le Puissance 4 est un jeu de stratégie dont les règles sont les suivantes : sur une grille verticale, chaque joueur viendra déposer à son tour un pion d'une couleur qui lui est propre. Chaque pion tombera donc au fond de la grille dans la colonne choisie, à moins qu'un pion déjà présent le bloque durant sa chute.

Le but du jeu est simple. Il s'agit ici d'une adaptation de ce jeu à l'Amstrad : la grille comporte huit cases verticales et huit horizontales. Le premier des deux joueurs

qui aura aligné quatre pions soit horizontalement, soit verticalement, soit en diagonale aura gagné.

Pour notre adaptation deux types de jeu seront possibles. Soit l'ordinateur remplacera simplement la grille et donc deux joueurs taperont à leur tour le numéro de la colonne où ils désirent lâcher leur pion, soit l'ordinateur sera votre adversaire. Dans ce dernier cas, il demandera souvent un délai de réflexion entre chaque coup, mais cela sera indiqué sur l'écran. Ces quelques points précisés, pas-

sons à l'écriture du programme.

En premier lieu nous demanderons à l'ordinateur de fonctionner en mode 1. Puis nous rédéfinirons l'ensemble des symboles graphiques nécessaires au bon déroulement du jeu.

Vient ensuite la page de présentation. Notons qu'un pion de chaque couleur, suivi des mentions VOUS et MOI, sera affiché pour vous rappeler votre couleur lors d'une partie contre la machine. La grille de jeu sera tracée à l'aide d'une sous-routine puis nous initialiserons les variables nécessaires tant à la page de présentation que pour le déroulement du jeu.

Dans cette phase d'initialisation, nous commencerons par mémoriser l'état d'occupation de l'ensem-

```

10 MODE 1: BORDER 1: PAPER 0: PEN 1
20 SYMBOL AFTER 127
30 SYMBOL 128,3,15,63,63,127,127,255,255
40 SYMBOL 129,192,240,252,252,254,254,255,255
50 SYMBOL 130,255,255,127,127,63,63,15,3
60 SYMBOL 131,255,255,254,254,252,252,240,192
70 ENV 1,1,15,1,1,-1,3,7,-2,1
80 ENV 2,3,5,1,1,0,1,1,-15,1
90 FOR I=1 TO 7
100 LET P(I,1)=1
110 NEXT I
120 REM *****
130 REM * PAGE DE PRESENTATION. *
140 REM *****
150 GOSUB 790
160 LOCATE 10,1: PRINT "PUISSANCE 4."
170 LET PAGE=1
180 LET C=1
190 LET EV=0
200 LET D=INT(RND*7)+1
210 IF C=1 THEN LET CP=5: GOSUB 980
220 IF C=-1 THEN LET CP=3: GOSUB 980
230 GOSUB 1120
240 IF EV=6 THEN GOSUB 1540
250 LET C=C*-1
260 LET K$=INKEY$
270 IF K$<>" " THEN LET PAGE=0
280 IF PAGE =1 THEN GOTO 200
290 REM *****
300 REM * DEBUT DE PARTIE; *
310 REM * INITIALISATION DES VARIABLES. *
320 REM *****
330 LET EV=0
340 LET DEB=0
350 LET C=0
360 LET CT=0
370 LET CO=0
380 LET CG=0
390 LET D=0
400 GOSUB 1540
410 PEN 5: LOCATE 25,16: PRINT CHR$(128);CHR$(129)
420 LOCATE 25,17: PRINT CHR$(130);CHR$(131)
430 PEN 3: LOCATE 25,20: PRINT CHR$(128);CHR$(129)
440 LOCATE 25,21: PRINT CHR$(130);CHR$(131)
450 PEN 2:LOCATE 30,17:PRINT "VOUS":LOCATE 30,21:
PRINT "MOI"
460 LOCATE 5,5:PRINT "VOULEZ-VOUS COMMENCER ? (O/N)
)"
470 LET K$=UPPER$(INKEY$):IF K$=" " THEN GOTO 470
480 LET C=-1:IF K$="0" THEN LET C=1
490 IF C=-1 THEN GOTO 580
500 PEN 2
510 LOCATE 5,5:PRINT "CHOISISSEZ VOTRE COLONNE SVP
)"
520 LET K$=INKEY$: IF K$=" " THEN GOTO 520
530 LET D=ASC(K$)-48:IF D<1 OR D>7 THEN GOTO 520
540 IF P(D,7)=0 THEN GOTO 570
550 LOCATE 5,5:PRINT "VOUS NE POUVEZ PAS JOUER ICI
!"
560 GOTO 520
570 LET DEB=DEB+1:LET CP=5
580 IF C=1 THEN GOTO 620

```

```

590 LET CP=3: PEN 2
600 LOCATE 5,5:PRINT "C'EST A MON TOUR DE JOUER ..
)"
610 GOSUB 1330
620 IF CP=3 THEN LOCATE 5,5:PRINT "ATTENDEZ, JE PR
EPARE MON COUP."
630 IF CP=5 THEN LOCATE 5,5:PRINT "MERCI; J'EXAMIN
E VOTRE CHOIX."
640 IF D=0 THEN LET D=INT(RND*7)+1:IF P(D,7)>0 TH
EN LET D=0:GOTO 640
650 GOSUB 1670
660 IF G=1 THEN GOTO 710
670 GOSUB 980:GOSUB 1120
680 IF G=1 THEN GOTO 710
690 LET C=C*-1
700 GOTO 490
710 IF GJ=4 THEN LOCATE 5,5:PRINT "BRAVO; VOUS GAG
NEZ LA PARTIE."
720 IF GO=4 THEN LOCATE 5,5:PRINT "DESOLE, C'EST M
OI QUI GAGNE!!!"
730 IF INKEY$=" " THEN GOTO 730
740 CLEAR:GOTO 10
750 REM *****
760 REM * SOUS ROUTINE REPRESENTATION *
770 REM * DE LA GRILLE DE JEU. *
780 REM *****
790 PAPER 2: PEN 0
800 FOR Y=12 TO 22 STEP 2
810 LOCATE 1,Y: PRINT " ": LOCATE 16,Y:PRINT " "
820 LOCATE 1,Y+1:PRINT " ": LOCATE 16,Y+1:PRINT " "
830 FOR X=2 TO 14 STEP 2
840 LOCATE X,Y: PRINT CHR$(128);CHR$(129)
850 LOCATE X,Y+1: PRINT CHR$(130);CHR$(131)
860 NEXT X
870 NEXT Y
880 FOR I=1 TO 16
890 LOCATE I,11: PRINT " "
900 NEXT I
910 LOCATE 1,24:PRINT " 1 2 3 4 5 6 7 "
920 PAPER 0: PEN 1
930 RETURN
940 REM *****
950 REM * SOUS ROUTINE D'ANIMATION DU *
960 REM * DEPLACEMENT HORIZONTAL DES PIONS. *
970 REM *****
980 PEN CP
990 FOR H=36 TO 2*D STEP-1
1000 LOCATE H,9: PRINT CHR$(128);CHR$(129);" "
1010 LOCATE H,10: PRINT CHR$(130);CHR$(131);" "
1020 SOUND 1,10+(3*H),0,0,2,0,0
1030 SOUND 1,20+(5*H),0,0,2,0,0
1040 NEXT H
1050 LOCATE H+1,9: PRINT " "
1060 LOCATE H+1,10: PRINT " "
1070 RETURN
1080 REM *****
1090 REM * SOUS ROUTINE D'ANIMATION DE *
1100 REM * LA CHUTE DES PIONS. *
1110 REM *****
1120 PAPER 2: LET Y=7
1130 LET XC=2*D
1140 LET YC=26-(2*Y)
1150 IF Y=7 THEN GOTO 1180

```

ble des cases de la grille en créant un tableau de mémorisation à l'aide de l'instruction DIM.

La variable C sera utilisée pour faire jouer chaque participant alternativement, PAGE indiquera à l'ordinateur si le jeu a débuté ou s'il est toujours sur la page de présentation, D sera utilisée pour déterminer la colonne où sera lâché le pion et CP mémoriserà sa couleur.

Au démarrage du programme le jeu rebouclera sur la page de présentation. Pour entamer la partie, il suffira de taper une touche. Dès lors, la grille sera vidée de son contenu et les variables initialisées. Ici interviendront quatre nouvelles variables CT, G, XP et CG. CT permet à la machine de savoir combien de

pions d'une même couleur sont alignés, G détecte l'alignement de quatre pions et indiquera donc la présence d'un gagnant, XP prendra une valeur correspondant à l'état d'occupation de la case sur laquelle porte l'analyse, et CG indiquera à l'ordinateur que placer l'un de ses pions sur cette case lui permettra de remporter la partie. Notons que DEB, ACC et EV seront utilisées pour réduire la durée de calcul de l'ordinateur en début de partie. DEB totalise le nombre de coups joués depuis le début de la partie ; si ce dernier est insuffisant pour qu'il soit possible que quatre pions puissent être alignés, les sous-routines de contrôle seront abrégées. De même, ACC est lié à DEB et confirme que pour les

lignes horizontales du tableau tout contrôle est inutile. EV vérifie le nombre de pions empilés sur la plus haute des colonnes. Si celui-ci reste inférieur à quatre, tout contrôle des verticales ou des diagonales est inutile.

En tout début de partie, l'ordinateur, lors du fonctionnement en mode automatique, propose au joueur de commencer ou non. En fonction de la réponse fournie, C prendra la valeur 1 ou 0, ce qui déterminera l'ordre de départ de la séquence d'alternance des coups. Ensuite, chaque fois que ce sera au joueur de lancer son pion, il lui faudra indiquer à l'ordinateur la colonne où il désire le lâcher en tapant son numéro au clavier.

Henri-Pierre Penel ▲

```

1160 PEN 0: LOCATE XC,YC-2: PRINT CHR$(128); CHR$(129)
1170 LOCATE XC,YC-1:PRINT CHR$(130);CHR$(131)
1180 PEN CP
1190 LOCATE XC,YC:PRINT CHR$(128);CHR$(129)
1200 LOCATE XC,YC+1:PRINT CHR$(130);CHR$(131)
1210 SOUND 1,10+(10*Y),0,0,1,0,0
1220 SOUND 2,200+(20*Y),0,0,1,0,0
1230 LET Y=Y-1
1240 IF P(D,Y)=0 THEN GOTO 1140
1250 LET P(D,Y+1)=CP
1260 IF EV<(Y+1) THEN LET EV=Y
1270 PAPER 0
1280 RETURN
1290 REM *****
1300 REM * SOUS ROUTINE D'ANALYSE DES *
1310 REM * ALIGNEMENTS HORIZONTAUX. *
1320 REM *****
1330 LET D=0: LET DL=0
1340 FOR Y=2 TO EV+1
1350 FOR H=1 TO 4
1360 IF P(H,Y)=0 AND P(H,Y-1)<>0 AND P(H+1,Y)=5
AND P(H+2,Y)=5 AND P(H+3,Y)=5 THEN LET DL=H
1370 IF P(H,Y)=0 AND P(H,Y-1)<>0 AND P(H+1,Y)=3
AND P(H+2,Y)=3 AND P(H+3,Y)=3 THEN LET DL=H
1380 IF P(H,Y)=5 AND P(H+1,Y)=0 AND P(H+1,Y-1)<>0
AND P(H+2,Y)=5 AND P(H+3,Y)=5 THEN LET DL=H+1
1390 IF P(H,Y)=3 AND P(H+1,Y)=0 AND P(H+1,Y-1) AND
P(H+2,Y)=3 AND P(H+3,Y)=3 THEN LET DL=H+1
1400 IF P(H,Y)=5 AND P(H+1,Y)=5 AND P(H+2,Y)=0 AND
P(H+2,Y-1)<>0 AND P(H+3,Y)=5 THEN LET DL=H+2
1410 IF P(H,Y)=3 AND P(H+1,Y)=3 AND P(H+2,Y)=0 AND
P(H+2,Y-1)<>0 AND P(H+3,Y)=3 THEN LET DL=H+2
1420 IF P(H,Y)=5 AND P(H+1,Y)=5 AND P(H+2,Y)=5 AND
P(H+3,Y)=0 AND P(H+3,Y-1)<>0 THEN LET DL=H+3
1430 IF P(H,Y)=3 AND P(H+1,Y)=3 AND P(H+2,Y)=3 AND
P(H+3,Y)=0 AND P(H+3,Y-1)<>0 THEN LET DL=H+3
1440 IF DL<>0 THEN LET H=4:LET Y=EV+1
1450 SOUND 1,10+(10*H),0,0,1,0,0
1460 SOUND 2,50+(10*Y),0,0,1,0,0
1470 NEXT H
1480 NEXT Y
1490 LET D=DL
1500 RETURN
1510 REM *****
1520 REM * SOUS ROUTINE EFFACEMENT GRILLE. *
1530 REM *****
1540 FOR XO=1 TO 7
1550 FOR YO=2 TO 7
1560 LET P(XO,YO)=0
1570 NEXT YO
1580 SOUND 1,25*XO,0,0,1,0,0
1590 SOUND 2,100/XO,0,0,1,0,0
1600 NEXT XO
1610 GOSUB 790
1620 RETURN
1630 REM *****
1640 REM * SOUS ROUTINE DE CONTROLE ET DE *
1650 REM * DECISION POSITION PION ORDINATEUR *
1660 REM *****
1670 LET GJ=0: LET GO=0
1680 FOR Y=2 TO EV+1
1690 LET GJ=0: LET GO=0
1700 FOR H=1 TO 7
1710 LET GJ=(GJ+((P(H,Y)=5)*-1))*((P(H,Y)<>5)+1)
1720 LET GO=(GO+((P(H,Y)=3)*-1))*((P(H,Y)<>3)+1)
1730 IF GJ=4 OR GO=4 THEN LET G=1: LET Y=EV+1: LET
H=7
1740 NEXT H
1750 NEXT Y
1760 IF GJ=4 OR GO=4 THEN LET G=1
1770 IF EV<3 OR G=1 THEN RETURN
1780 LET GJ=0: LET GO=0
1790 LET DC=0
1800 FOR H=1 TO 7
1810 LET GJ=0:LET GO=0
1820 FOR Y=2 TO EV+1
1830 LET GO=(GO+((P(H,Y)=3)*-1))*((P(H,Y)<>3)+1)
1840 LET GJ=(GJ+((P(H,Y)=5)*-1))*((P(H,Y)<>5)+1)
1850 SOUND 1,50+(10*GJ),0,0,1,0,0
1860 SOUND 2,60+(10*GO),0,0,1,0,0
1870 IF GJ=4 OR GO=4 THEN LET G=1: LET Y=EV+1: LET
H=7
1880 IF Y=7 THEN LET DC=0: GOTO 1920
1890 IF P(H,Y+1)<>0 THEN GOTO 1920
1900 IF GO=3 THEN LET DC=H
1910 IF DC=0 AND GJ=3 THEN LET DC=H
1920 NEXT Y
1930 NEXT H
1940 IF G=1 THEN RETURN
1950 LET GJ=0: LET GO=0
1960 LET DD=0
1970 FOR Y=2 TO 4
1980 FOR H=1 TO 4
1990 SOUND 1,50+(10*GO),0,0,1,0,0
2000 SOUND 2,60+(10*GJ),0,0,1,0,0
2010 IF P(H,Y)=5 AND P(H+1,Y+1)=5 AND P(H+2,Y+2)=5
THEN LET GJ=3: IF GJ=3 AND P(H+3,Y+3)=5 THEN LET
GJ=4
2020 IF GJ=3 AND P(H+3,Y+3)=0 AND P(H+3,Y+2)<>0
THEN LET DD=H
2030 IF P(H+3,Y)=5 AND P(H+2,Y+1)=5 AND P(H+1,Y+2)
=5 THEN LET GJ=3: IF GJ=3 AND P(H,Y+3)=5 THEN LET
GJ=4:LOCATE 1,3
2040 IF GJ=3 AND P(H,Y+3)=0 AND P(H,Y+2)<>0 THEN
LET DD=H
2050 IF P(H,Y)=3 AND P(H+1,Y+1)=3 AND P(H+2,Y+2)=3
THEN LET GO=3: IF GO=3 AND P(H+3,Y+3)=3 THEN LET
GO=4
2060 IF GO=3 AND P(H+3,Y+3)=0 AND P(H+3,Y+2)<>0
THEN LET DD=H
2070 IF P(H+3,Y)=3 AND P(H+2,Y+1)=3 AND P(H+1,Y+2)
=3 THEN LET GO=3: IF GO=3 AND P(H,Y+3)=3 THEN LET
GO=4
2080 IF GO=3 AND P(H,Y+3)=0 AND P(H,Y+2)<>0 THEN
LET DD=H
2090 IF GJ=4 OR GO=4 THEN LET G=1:LET H=4:LET Y=4
2100 NEXT H
2110 NEXT Y
2120 IF G=1 THEN RETURN
2130 IF Y=7 THEN RETURN
2140 IF DD<>0 THEN LET D=DD
2150 IF DC<>0 THEN LET D=DC
2160 RETURN

```